

You can always make more primes

To refute the conjecture that there are only finitely many primes, we need merely construct an infinite set of “primoids”, coprime integers which may or may not be prime. Even if we don’t bother to factor them, the union of all their factors will be infinite, and in fact, disjoint.

We construct the primoid infinite set by starting with {}. Multiply “them” all together (getting 1, the empty product). Then add 1, and iterate:

```
In[3]:= Nest[Append[#, 1 + Times @@ #] &, {}, 8] // Echo // PrimeQ
>> {2, 3, 7, 43, 1807, 3 263 443, 10 650 056 950 807, 113 423 713 055 421 844 361 000 443}
Out[3]= {True, True, True, True, False, True, False, False}
```

(It so happens that five of these primoids are actually prime. Note for later the ...807, ...443 repetition.) The primoids are defined by this “full history” recurrence:

Out[10]=
$$p(n+1) = \prod_k^n p(k) + 1$$

But this is easily converted to a first order recurrence! You may wish to pause here and discover it yourself.

```
In[47]:= %10 /. n -> n - 1
Out[47]= 
$$p(n) = \prod_k^{n-1} p(k) + 1$$

```

```
In[48]:= detour[%10, n - 1]
Out[48]= 
$$p(n+1) = p(n) \prod_k^{n-1} p(k) + 1$$

```

(`detour` is my own handy function for breaking the range of a sum or product in two:)

```
detour[xp_, d_] :=
  xp /. (op : Sum | Product | Inactive[Sum | Product]) [a_, {v_, L___, h_}] =>
    (op /. {(Sum | Inactive[Sum]) -> Plus, Product | Inactive[Product] -> Times}) [
      op [a, {v, L, d}], op [a, {v, d + 1, h}] ]
```

Now we can eliminate the common $\prod_k^{n-1}$ expression:

```
In[51]:= Eliminate[{%47, %48}, %47[[2, 2]]]
Out[51]= 
$$p(n+1) = p(n)^2 - p(n) + 1$$

```

a first order recurrence.
And now, a curiosity. Primoids loop mod 10^{18} with period 65536:

```
NestList[Nest[Mod[1 - # + # #, 10 ^ 18] &, #, 2 ^ 16] &, 2, 3] // tim
0.350538,4
Out[42]= {2, 847 003 581 666 295 807, 847 003 581 666 295 807, 847 003 581 666 295 807}
```

What is this number ...847003581666295807? Raise it to the 5th power mod 10^{18} :

```
In[43]:= Mod[%[[-1]] ^ 5, 10 ^ 18]
Out[43]= 847 003 581 666 295 807
```

OEIS strikes again!